

3. Selected SAS code

Input variables

&xi = macro variable representing the number of events/cases (the numerator)
&mi = macro variable representing the at risk population (the denominator)
&outcome = macro variable containing a label for the outcome (type of event/case)
code = unique identifier for each study

* calculation proportion and 95% confidence interval (and their logit transforms) for each study;

```
data contact1;
set contacts;
xi=&xi;
mi=&mi;
z=1.96;
if mi = 0 then do;
    p=.;
    l95_p=.;
    u95_p=.;
end;
else do;
    p=xi/mi;
    l95_p=((2*mi*p+z**2-1)-z*sqrt(z**2-(2+1/mi)+4*p*(mi*(1-p)+1)))/(2*(mi+z**2));
    u95_p=((2*mi*p+z**2+1)+z*sqrt(z**2+(2-1/mi)+4*p*(mi*(1-p)-1)))/(2*(mi+z**2));
end;
drop z;
tlnprev_calc=log((xi+0.5)/(mi-xi+0.5));
var_lnprev_calc = 1/(xi+0.5)+1/(mi-xi+0.5);
run;
```

* estimate starting parameter values;

```
proc means data = contact1 noprint;
var tlnprev_calc ;
output out=parmsest (drop=_type__freq_) mean=mtlnprev var= vtlnprev;
run;
```

* fit binomial model with random intercept;

```
ods output Nlmixed.ParameterEstimates=param (keep=parameter estimate lower upper)
           Nlmixed.Dimensions=dimensions (where=(descr="Subjects"));
proc nlmixed data=contact1 (where=(xi<=mi)) df = 1000;
parms / data=parmsest; /*Initial values*/
pi = 1/(1+exp(-tlnprev));
model xi~binomial(mi,pi);
random tlnprev ~ normal(mtlprev , vtlnprev) subject=code;
run;
```

* back transform the resulting estimates;

```
data param1;
length parameter $15. outcome $32. ;
set param (where=(parameter='mtlnprev'));
array estim{3}estimate upper lower;
array estim1{3}estimate1 upper1 lower1;
do i = 1 to 3;
estim1{i}=1/(1+exp(-estim{i}));
end;
drop i upper lower;
outcome = "&outcome";
rename estimate1=proportion upper1=upper lower1=lower estimate=mtlnprev;
run;
```

* merge resulting datasets and then calculate Cochran's Q, tau2 and I2;

```
data contact1; set contact1; match=1; run;
data param1; set param1; match=1; run;
data param; set param; match=1; run;
```

```
data contact1;
merge contact1 (where=(&mi ne . and &xi ne .))
param1 (keep = match mtlprev proportion) param(where=(parameter='vtlnprev') keep= match parameter estimate);
by match;
wi=1/var_lprev_calc;
Qi=wi*(tlprev_calc - mtlprev)**2;
rename estimate=tau2;
drop match parameter;
run;
```

```
proc means data=contact1 noprint;
var qi;
output out=isquare sum=q n=k;
run;
```

```
data isquare;
set isquare (drop= _type_ _freq_);
i2 = 100*(q-(k-1))/q;
if i2 < 0 then i2=0;
if k = 1 then do;
    q = .;
    i2 = .; * set I2 and Q to missing if only one study;
end;
run;
```